


# 3D Konstruktion und 3D Druck mit OpenSCAD und Slic3r

## Einführung, Tutorial und Übungen

Prof. Dipl.-Ing. Klaus Knopper <klaus.knopper@hs-kl.de>



Arbeitsprozesse bei der Verarbeitung von 3D-Daten (Beispiel)

# 1 Über OpenSCAD

OpenSCAD ist eine IDE (Integrated Development Environment = Integrierte Programmierumgebung) und ein CAD-Programm (Computer Aided Design), das die Konstruktion von druckbaren 3D-Objekten, z.B. Bauteilen, in einer einfachen, geometrieorientierten Programmiersprache erlaubt.


**<http://www.openscad.org>**

Bereits mit wenigen Befehlen können komplexe Objekte erstellt werden. Maus bzw. Touchscreen werden dabei lediglich im Betrachtungsmodus verwendet, um die Objekte zu drehen (linke Maustaste), zu verschieben (mittlere Maustaste oder linke+rechte Maustaste) und zu zoomen (Mausrad).


OpenSCAD verwendet eine ähnliche Syntax wie C oder JAVA, und eignet sich daher auch gut als erster Einstieg in die Programmierung, zumal hier mit deutlich weniger Programmieraufwand schnelle Erfolgserlebnisse (beinahe) garantiert sind.

Eine sehr ähnliche Sprache mit anderem Fokus ist x3dom, was als HTML+XML-Dialekt im Web-Browser mit WebGL visualisierte Grafik- und Audio-Objekte produziert.

Für die Ausgabe der in OpenSCAD entworfenen Modelle auf einen 3D-Drucker ist aktuell das Standard-Dateiformat STL üblich. Von OpenSCAD als STL exportierte Modelle können in Software für 3D-Drucker importiert werden, z.B. in Slicer-Programme wie slic3r, und somit in den zur Druckersteuerung verwendeten G-CODE weiterverarbeitet werden (s.a. nachfolgende Abbildungen).



STL-Objekte in Filamentbahnen (G-CODE) umrechnen mit Slic3r



G-CODE drucken auf Teebot

## 2 OpenSCAD Installation

OpenSCAD wird als Open Source Programm unter den Bedingungen der GNU General Public License Version 2 entwickelt und verbreitet, die Nutzung und Weitergabe des Programms in unveränderter oder abgewandelter Form sowie die Integration in eigene Projekte sind hierdurch kostenlos möglich.

Fertig übersetzte, installierbare Versionen von OpenSCAD werden über die Webseite <http://www.openscad.org/downloads.html> angeboten.

Neben der schnellen, nativen Version von OpenSCAD, die direkt unter Linux, Windows oder Mac ausführbar ist, gibt es noch eine Browser/WebGL-basierte Variante mit leicht unterschiedlicher nativer Syntax: <http://openjscad.org/>


Mit OpenSCAD erstellte Dateien können per „Drag & Drop“ in OpenJSCAD im Browser dargestellt und getestet werden. Allerdings stehen nicht alle aktuellen OpenSCAD-Funktionen in OpenJSCAD zur Verfügung, daher sollte eher die native (Betriebssystemspezifische) Variante des Programms im Kurs verwendet werden.

Für Android auf Smartphones ist OpenSCAD mit etwas reduziertem Funktionsumfang als App unter dem Namen ScorchCAD verfügbar.

## 3 Bedienung

Nach dem Start von OpenSCAD wird typischerweise ein dreigeteiltes Fenster angezeigt:

1. Ein einfacher Texteditor für die Bearbeitung des Programms (Eingabe),
2. Ein Bereich für die graphische Ausgabe zum Betrachten des Ergebnisses (Ausgabe),
3. Ein Bereich, in dem Fehler- und Statusmeldungen angezeigt werden (Verarbeitung).



Neben der normalen Texteingabe im Editor stehen folgende Programm-Funktionen zur Verfügung:

<b>Editor-Bereich</b>	
<b>Interaktion</b>	<b>Wirkung</b>
F5	Übersetzung (Compile) und Gittermodell aktualisieren (schnelle Variante)
F6	Übersetzung (Compile) und aufwändige Version rendern (langsame Variante, für STL-Export notwendig)
F4	Dokument neu laden, sonst wie F5 (bei externem Quelltexteditor verwenden)
<b>Grafik-Bereich</b>	
<b>Interaktion</b>	<b>Wirkung</b>
Linke Maustaste	Modell drehen
Rechte Maustaste	Modell verschieben
Mausrad	Zoom

## 4 Syntax und Semantik der Programmiersprache OpenSCAD

Die Syntaxbeschreibung von OpenSCAD passt gut lesbar noch auf eine Din A4 Seite. Die kürzeste Variante zum Nachschlagen ist das aktuelle Online-Cheat-sheet:

**<http://www.openscad.org/cheatsheet/>**

Achtung: Die auf der Seite angegebene Syntaxbeschreibung ist wörtlich (literal), die verwendeten eckigen Klammern für die Angabe von Punkten oder Vektoren werden also wirklich genau so im Programmcode verwendet, und sind KEIN EBNF!

Die ausführliche, englische Beschreibung der einzelnen Funktionen (Semantik) von OpenSCAD ist unter

**[https://en.wikibooks.org/wiki/OpenSCAD\\_User\\_Manual](https://en.wikibooks.org/wiki/OpenSCAD_User_Manual)**

als Wikibook verfügbar. Einige der Beispiele und Abbildungen des Tutorials sind hieraus übernommen.

## 5 OpenSCAD Tutorial

### 5.1 Kommentare und Kommentarblöcke

In OpenSCAD können im Quelltext, wie bei den meisten anderen Programmiersprachen (z.B. JAVA, C++) Hinweise und Erklärungen untergebracht werden, die vom Compiler ignoriert werden. Man bezeichnet dies als **Kommentar**.

Hierfür gibt es zwei Varianten:

```
// Kommentar
```

gilt nur bis zum Ende der Zeile (benötigt also keine weitere Anweisung zur „Aufhebung“ des Kommentars), während

```
/* Kurzer Kommentar */
```

```
/* Hier sind  
mehrere  
Kommentarzeilen  
vorhanden */
```

einen **Kommentarblock** darstellt, der mit der Zeichenfolge `/*` beginnt, und mit der Zeichenfolge `*/` endet.

Kommentare können auch verwendet werden, um Teile des Quelltextes zu deaktivieren, man bezeichnet dies als **Auskommentieren**.

```
intersection() {  
  cube(60, center=true);  
  sphere(40);  
  /* cylinder(80,40); */ /* Später! */  
}
```

### 5.2 Anweisungen und Anweisungsblöcke

Anweisungen *müssen* wie in JAVA mit einem Semikolon (;) beendet werden.

```
Anweisung1; Anweisung2;
```

Anweisungen können z.B. Aufrufe primitiver Grafikfunktionen wie `cube()` sein.

Mehrere Anweisungen können (wie in JAVA) mit „geschweiften Klammern“ `{...}` zu einem *Anweisungsblock* zusammengefasst werden.


```
{  
  Anweisung1;  
  Anweisung2;  
  Anweisung3;  
}
```

Ein weiteres Semikolon hinter der schließenden Klammer ist (bis auf wenige Ausnahmen) nicht erforderlich, aber erlaubt.


### 5.3 Grafik-Primitiva

Primitive Grafikfunktionen zeichnen geometrische Objekte, deren Eigenschaften optional in den runden Klammern angegeben werden können. Dabei sind die Standardeinstellungen je nach Funktion unterschiedlich.

Während



einen Würfel mit Kantenlänge 1 zeichnet, erzeugt




einen Quader mit den Seitenlängen  $x = 2, y = 3, z = 4$ .

Die **positionalen Parameter** in der eckigen Klammer haben hier also die Bedeutung der drei Seitenlängen.

Beim `cube ( )` befindet sich die „linke untere Ecke“ im Mittelpunkt des Koordinatensystems. Der Würfel lässt sich mit


```
cube (center=true) ;
```



so verschieben, dass der Mittelpunkt des Koordinatensystems in seinem eigenen Mittelpunkt liegt, was dem Standard-Verhalten der Kugel entspricht (diese ist grundsätzlich an ihrem Mittelpunkt zentriert).


Die Funktion

```
sphere () ;
```



erzeugt eine Kugel mit Radius 1, die im Grafikfenster allerdings nicht wirklich wie eine Kugel aussieht. Dies liegt an der Standard-Auflösung, mit der Objekte durch Aneinanderlegen von Dreiecken dargestellt werden. Durch Setzen der Variable **\$fn=40**; kann die Auflösung verbessert werden, hierdurch werden mehr Dreiecke pro Zeicheneinheit gerendert, was sich vorwiegend bei runden Elementen lohnt. Allerdings erhöht sich dadurch die Rechenzeit für die Berechnung der Modelle und auch für die Darstellung der Objekte teilweise dramatisch.

```
$fn=40 ;  
sphere () ;
```




Objekte können zu einer neuen Form addiert werden, indem mehrere Anweisungen mit der Funktion

```
union () { anweisung1; anweisung2; ... }
```

zusammengefasst, oder einfach nur hintereinander aufgeschrieben werden:




```
sphere (40) ;  
cube (60, center=true) ;
```



In diesem Beispiel hat die Kugel den **Radius 40** und der zentrierte Würfel die **Seitenlänge 60**. Das gleiche Bild entsteht, wenn bei der Kugel der **Durchmesser 80** angegeben wird: **sphere (d=80) ;**.

Zylinder:

```
cylinder (r1=25, r2=0, h=50) ;
```



### Definition von Körpern (Polyeder)

Während ein **Polygon** ein zweidimensionales Objekt ist, wird ein **Polyeder** in der Mathematik oft als die „Außenpunkte“ mit  $(x, y, z)$  definiert. Leider ist diese rein Punktorientierte Darstellung für 3D-Volumenobjekte suboptimal, da nicht eindeutig ist, wie die Punkte miteinander verbunden werden, bzw. welche Außenflächen durch diese dargestellt werden, denn nicht immer ist der kürzeste Abstand zwischen zwei Punkten auch eine physikalische Kante.

Daher verwendet OpenSCAD eine Definition, die eher auf **Flächen** (faces) abzielt, welche durch die Angabe ihrer Eckpunkte, die alle in einer Ebene liegen müssen, dargestellt werden.

Das folgende Beispiel entstammt dem **OpenSCAD-Wiki zur polyhedron ()** - Funktion.

Zunächst werden Punkte im Raum als Array festgelegt (3D-Koordinaten):

```
Points = [  
  [ 0, 0, 0 ], //0  
  [ 10, 0, 0 ], //1  
  [ 10, 7, 0 ], //2  
  [ 0, 7, 0 ], //3
```


```
[ 0, 0, 5 ], //4
[ 10, 0, 5 ], //5
[ 10, 7, 5 ], //6
[ 0, 7, 5 ]]; //7
```

Nun werden alle Flächen des Körpers festgelegt, wobei sich die Zahlenangaben jetzt auf den Index der Eckpunkte im zuvor definierten **Points**-Array beziehen:

```
Faces = [
  [0,1,2,3], // bottom
  [4,5,1,0], // front
  [7,6,5,4], // top
  [5,6,2,1], // right
  [6,7,3,2], // back
  [7,4,0,3]]; // left
```

Jetzt kann die **polyhedron()**-Funktion aufgerufen werden, um den Polyeder darzustellen (in diesem Fall ein einfacher Quader):

```
polyhedron(Points, Faces);
```



Weitere Grafikobjekte und die zugehörigen Parameter im „Live-Test“ (s.a. <http://www.openscad.org/cheatsheet/>).

## 5.4 Einfache Transformationen

Alleine mit der Addition von Objekten lassen sich noch keine beliebigen 3D-Formen herstellen, hierzu sind Verknüpfungs- sowie Positionierungsfunktionen notwendig, sogenannte **Transformationen**.


In OpenSCAD bezieht sich eine Transformation immer auf die direkt dahinter stehende Anweisungsfolge, die Reihenfolge der Anweisungen ist also eigentlich „von rechts nach links“ zu lesen.

Im folgenden Beispiel wird der Würfel aus dem vorigen Beispiel um 10 Einheiten in x-Richtung (nach rechts) verschoben.

```

sphere (40) ;
translate ([10, 0, 0])
  cube (60, center=true) ;

```




Hier werden hingegen **beide** Objekte gemeinsam um 70 Einheiten nach rechts verschoben, was durch die „Klammerung“ erreicht wird:

```

translate ([70, 0, 0]) {
  sphere (40) ;
  cube (60, center=true) ;
}

```




Hier wird ein zweiter Würfel an der X-Achse um 45 Grad (positiv = links) gekippt:

```

color ("green") cube (40) ;
rotate ([45, 0, 0]) cube (40) ;

```



Vorsicht: Eine Translation oder Rotation wird immer vom Mittelpunkt des Koordinatensystems aus durchgeführt, und *nicht* vom Mittelpunkt des Objektes aus, wie man vielleicht annehmen würde!

Bei einer Anweisungsfolge ist entscheidend, in welcher Reihenfolge welches Kommando ausgeführt wird. Wird die Reihenfolge verändert, entstehen völlig andere Szenarien oder Objekte, auch bei ansonsten gleichen Parametern.


Weitere Beispiele (Demo): Rotieren, Skalieren, Einfärben, Spiegeln

Eine verschachtelte Anweisungsfolge (die Werte in eckigen Klammern sind jeweils Längen in x,y,z-Richtung, bei der Rotation Winkel um x,y,z-Achse):

```

mirror([1,1,0])
translate([0,0,52]) {
  translate([0,0,-52])
  // relativ skalieren
  scale([1.5,1.5,1.5])
  // absolut skalieren
  // resize([50,50,100])
  color("green") sphere(25);
  rotate(45,[0,0,1]) cube(25);
}

```




Nur die **Schnittmenge** von Objekten darstellen:

```

intersection() {
  sphere(20);
  cube(30, center=true);
}

```




Nur die **Differenz** von Objekten darstellen, d.h. vom ersten Objekt alle weiteren Objekte „abziehen“.

```

difference() {
  cube(30, center=true);
  sphere(20);
}

```




Tipp: Mit Präfix # vor einer Anweisung kann deren Inhalt „halbdurchsichtig“ dargestellt werden, also z.B. so:

```

difference() {
  cube(30, center=true);
  #sphere(20);
}


```




## 5.5 Mehrfache Ausführung (Schleifen)

Bei diesem Konstrukt wird eine Anweisungsfolge mehrfach, mit sich verändernden „Laufvariablen“ ausgeführt. Die Syntax ist hier bei OpenSCAD kompakter, aber leider ausnahmsweise etwas weniger eingängig als in der aus PHP, JAVA oder C bekannten `for()`-Schleife.

```
for(x=[0:3]){
  for(y=[0:3]){
    translate([x*25,y*25,0])
    cube(20);
  }
}
```



```
for(x=[0:3], y=[0:3], z=[0:3]) {
  translate([x*25,y*25,z*25]) {
    if( (x*y*z)%2 == 0) {
      cube(20);
    }
  }
}
```




## 5.6 Komplexe Transformationen

Einige Transformationen sind sehr rechenaufwändig.


**minkowski**: Kanten abrunden (bzw. zweites Objekt an den Kanten des ersten anwenden). Achtung: Hierdurch vergrößert sich das Objekt um den Radius der für die Abrundung verwendeten Kugel auf jeder Seite!

```
$fn=20;
minkowski() {
  cube(25);
  sphere(5);
}
```




**hull**: Eine „minimale“ Hülle (Folieneffekt) um Objekte zeichnen.

```
$fn = 20;  
hull() {  
  #translate([0,0,0])  
  cylinder(r=2,h=2);  
  #translate([10,0,0])  
  cylinder(r=2,h=2);  
  #translate([5,10,0])  
  cylinder(r=2,h=2);  
}
```




**rotate\_extrude**: Rotationskörper definieren.

```
$fn=60;  
rotate_extrude() {  
  translate([50,20]) circle(20);  
}
```



**linear\_extrude**: Flächen in eine Richtung „herausziehen“ und (optional) dabei verdrehen.

```
$fn=80;  
linear_extrude(height=50,  
center=true, twist=120,  
slices=40) {  
  hull() {  
    translate([10,0])  
    circle(15);  
    rotate(45)  
    square(23,center=true);  
  }  
}
```




## 5.7 Von 2D nach 3D

### Text extrudieren


Wie `circle()` oder `square()` ist `text()` eine Funktion im zweidimensionalen Raum und besitzt definitionsgemäß keine Ausdehnung in Z-Richtung. Solche Objekte sind noch nicht direkt druckbar.

Um hieraus die bekannten 3D-„Schlüsselanhänger“ (s. Übung) zu erzeugen, kann ein Text-Objekt mit Hilfe der Funktion `linear_extrude()` in Z-Richtung aus der Ebene „herausgezogen“ werden.

```
text("Hallo, Welt!",  
    size = 20,  
    font = "SansSerif:style=Bold",  
    halign = "center",  
    valign = "center",  
    $fn = 40);
```



```
linear_extrude(height = 5)  
text("Hallo, Welt!",  
    size = 20,  
    font = "SansSerif:style=Bold",  
    halign = "center",  
    valign = "center",  
    $fn = 40);
```



### Graustufen-Bilder extrudieren

Die seit OpenSCAD Version 2015.03 verfügbare Funktion `surface()` erlaubt es, Textdateien mit Höhenkoordinaten oder Graustufen-Bilder im PNG-Format als Oberflächen-Relief darzustellen.



```
surface(file = "smiley.png",  
    center = true);
```






```
surface(file = "smiley.png",  
center = true,  
invert=true);
```



## 5.8 3D-Dateien importieren

OpenSCAD unterstützt, abhängig von aktivierten Erweiterungen, eine Anzahl von „Fremdformaten“ wie STL oder DXF, die als Vektorgrafik importiert und durch eigene Konstrukte erweitert werden können.

```
// Sockel-Profil aus DXF-Datei  
scale(0.2)  
translate([0, 0, 30])  
rotate_extrude(convexity = 10)  
import_dxf("sockel_profil.dxf");  
// Kopf aus STL-Datei  
translate([-8, -12, 28])  
scale(3.2)  
import("horse3.stl");
```



Auf diese Weise lässt sich beispielsweise auch Stützmaterial für Objekte mit Überhang einfügen, die nicht direkt in OpenSCAD erstellt wurden.

## 5.9 Eigene Funktionen (Module) erstellen

Module (analog JAVA: Methoden) erweitern den Sprachumfang von OpenSCAD durch Implementierung von Algorithmen, die sich aus bereits bekannten Anweisungen zusammensetzen.

```
Syntax: module name_des_moduls (parameter1=standardwert1, ...) {  
  Anweisungen ...  
}
```

Später wird das Modul wie die vordefinierten Funktionen verwendet:

```
name_des_moduls(10, 20, ...);
```




```

module spaceball( radius=15,
                  seite=20 ) {
  sphere(radius);
  for(rx=[0:7], ry=[0:7], rz=[0:7])
    rotate([rx*45, ry*45, rz*45])
      cube(seite, center=true);
}

spaceball();

```



Module, die analog `translate()` auf die nachfolgenden Anweisungen wirken, können mit der Funktion `child()`; auf die zuvor produzierten Objekte zugreifen.

```


module mirror_and_show(vec=[1,0,0]) {
  mirror(vec) child();
  // # = Objekt transparent anzeigen
  // % = Objekt transparent nur in der Ausgabe anzeigen, nicht drucken
  %child();
}

```


## 6 Übungen

6.1 Stellen Sie die abgebildeten Objekt in OpenSCAD dar.


6.1.1 Spielfigur



6.1.2 Rad mit Aussparungen




6.1.3 Tetraeder mit abgerundeten Kanten



Hinweis: In diesem Beispiel war `minkowski ()` nicht erforderlich.


#### 6.1.4 Podest, innen hohl (um Material zu sparen)



#### 6.1.5 Schlüsselanhänger als Modul (!) mit als Parameter wählbarer Beschriftung




#### 6.1.6 Eine Gewindestange




Hinweis: In diesem Beispiel wurde weder `rotate_extrude()`, noch `linear_extrude()` verwendet.

### 6.1.7 Eine Pyramide



**6.2 Erstellen Sie in OpenSCAD eine parametrische Funktion (d.h. ein OpenSCAD-Modul, das Übergabeparameter verarbeiten kann), welche ein bewegliches Scharnier (Rotationsgelenk) in der abgebildeten Form generiert. Die Übergabeparameter sollen die Länge, Breite und Höhe des Objekts festlegen, zwischen den kegelförmigen Aussparungen bzw. Spitzen soll der Abstand aber in jedem Fall genau 0,3 Millimeter betragen, damit das Gelenk nicht im Drucker „verklebt“ und genügend Spiel für die Bewegung bleibt.**



## 7 Links

<http://www.openscad.org/> OpenSCAD Homepage

<http://www.openscad.org/cheatsheet/> OpenSCAD „Cheat Sheet“ - die ultrakurze Kompakt-Beschreibung aller OpenSCAD-Funktionen (mit Links zu Details)

<http://www.thingiverse.com/> Druckbare 3D-Objekte, Beispiele und Experimente zum Herunterladen, teilweise auch mit Quelltexten für OpenSCAD

<http://slic3r.org/> Slic3r, ein Slicer (Ebenenzerlegung, Perimeter und Füllung berechnen und G-CODE für Drucker erzeugen) für 3D-Formate

<http://www.octoprint.org/> OctoPrint, interaktive web-basierte Steuerung und Überwachung von 3D-Druckern auf Basis eines Raspberry Pi2 Mini-computers

<https://www.hs-kl.de/betriebswirtschaft/studiengaenge/bachelor/> Studiengang Information Management an der Hochschule Kaiserslautern, Campus Zweibrücken

<http://knopper.net/tuebix/> Unterlagen zu den TÜBIX Vorträgen von Klaus Knopper